
SHORT MAPLE MANUAL

Algebraic Operations, Functions and Constants
Elementary algebraic operations are denoted by +,-,* and /. The sign for exponentiation
is ^. Only one type of parentheses is allowed, namely the round parentheses ( ). If a
parenthesis does not have its mathcing pair Maple will generate an error message.

The symbol for the absolute value is abs(). For example, the expressions |x − 6| and
||f(x)| − |x|| have to be typed as > abs(x-6); and > abs(abs(f(x))-abs(x));

The symbol ! is used for factorials. To find the numeric value of 1000!, type > 100!;

Constants: π is denoted by Pi (uppercase P) and e, the basis of natural logarithms, is
obtained by executing the command exp(1). To type ∞ use infinity.

The symbol for powers is ^. To type x3, t6/7 and z−2.11 use x^3, t^(6/7) and z^(-
2.11). The square root function

√
x can be input as sqrt(x) or as x^(1/2). The symbol

for n
√

x is surd(x,n). For example, 3
√

2x + 4 5
√

x2 is entered as

> surd(2x,3)+4*surd(x^2,5);

or, of course, as

> (2x)^(1/3)+4*(x^2)^(1/5);

The symbols for trigonometric functions are sin(x), cos(x), tan(x), cot(x), sec(x)
and csc(x). The variable x is in radians. Inverse trigonometric functions are denoted by
arcsin(x), arctan(x), etc. The exponential function ex is denoted by exp(x) and its
inverse, the natural logarithm function lnx by ln(x). The symbols for hyperbolic functions
are sinh(x), cosh(x), tanh(x), etc.

The symbol % refers to the result of the previous computation. For example,

> 4*5;

will return 20; the command

> %/10;

will return 2 (since % is equal to 20). But now

> %/10;

will return 1/5, since % is equal to 2.

Evaluating Numeric Expressions
Maple always tries to return the answer in the most accurate form. That is the reason
why, for example, we see the response arctan(2) to the command

> solve(tan(x)=2,x);

that asks for the solution of the equation tanx = 2. To obtain a decimal (numeric) approx-
imation of arctan 2, we use either > evalf(%); immediately after the solve command or
combine the two commands into one: > evalf(solve(tan(x)=2,x)); The last command
can be abbreviated as > fsolve(tan(x)=2,x);

If more digits are needed, we use

evalf(expression,number of digits)

The default value is 10. To display the first 45 digits of the number 223/115, we use

> 2^23/11^5;evalf(%,45);

or, joined together in a single command

> evalf(2^23/11^5,45);

The command > evalf(exp(1),1000); will return 1000 digits (hence 999 decimals) of
the basis e = 2.71828... of natural logarithms.

Maple will sometimes respond with a symbolic expression, although we have assigned
numeric values to all symbols involved. If this is the case, execute the command evalf(%)
immediately after the response that should have been evaluated. For example, after defin-
ing the function

> f := x -> exp(x)+sin(x)+x^2+1;

we ask for its value at x = 2 :

> subs(x=2,f(x));

but the response e2 + sin 2 + 5 is probably not what we were looking for. To obtain a
numeric value, we execute > evalf(%); immediately after subs command. Alternatively,
use > evalf(subs(x=2,f(x)));

To simplify the symbolic expression we have obtained, we use the command

simplify(expression)

where expression can be replaced by % if we want to tell Maple to simplify the result we
have just obtained.

Assigning Values
The command

symbol := value

assigns the value to the symbol. For example, > a := 23; assigns the numeric value 23
to the constant a. The assignment

> m := fsolve(4*m^3-11=0,m);

assigns the decimal approximation 1.401019665 of the solution of the equation 4m3−11 = 0
to the symbol m.

Functions
The command

function := variable -> rule

defines a function: function is the name of the function and variable defines its variable.

For example, to define a function f(x) = 3x − tanx, we use

> f := x -> 3*x-tan(x);

The command > h := u -> (u^2-u)/(u+u^3-1); defines the function h(u) =
u2 − u

u + u3 − 1
To find the value of f(x) at x = 2, we can use either

> evalf(f(2));

or the subs command: > subs(x=2,f(x));

Likewise, to compute the value of the previously defined function h(u) at u = −0.2, we
can use either > evalf(h(-0.2)); or > evalf(subs(u=-0.2,h(u))); (remember that
we have to use evalf command to ask for a decimal number as a result; without it, Maple
might return some algebraic expression or will just retype our command).

Every time we refer to a function, we have to keep the independent variable, i.e., we
need f(x) and not just f. Functions can be used to define other functions. We have
already defined f(x); now, to define the function g(x) = 2f(x)2 − cos(f(x)), we execute
the command

> g := x -> 2*f(x)^2-cos(f(x));

Notice that we always use f(x) (and not just f) when refering to a function.
subs command becomes useful when one has to compute the value of a derivative of

a function. For example, to compute f ′(3.1) we use

> evalf(subs(x=3.1,diff(f(x),x)));

and to compute f ′′(3.1) we use

> evalf(subs(x=3.1,diff(f(x),x$2)));

Plotting Functions
To plot the graph of a function, use

plot(function,variable=interval)

where interval defines the interval where the plot will be constructed. For example, to
obtain the plot of the function f(x) = x2 sin x − cos(3x) on the interval [0, π] execute

> plot(x^2*sin(x)-cos(3*x),x=0..Pi);

To see the plot of h(s) = arcsin(s2) + s on [0, 1], you can either execute

> plot(arcsin(s^2)+s,s=0..1);

or, define the function first

> h := s -> arcsin(s^2)+s;

and then plot it

> plot(h(s),s=0..1);

The command

> plot(diff(h(s),s),s=0..1);

will plot the graph of h′(s) over [0, 1].

The way Maple draws the graph is the following: it plots a lot of points (as we would
do: by choosing x and computing the corresponding value y = f(x)) and then connects
them with line segments. The command
> plot(x^2*sin(x)-cos(3*x),x=0..Pi,style=point);

will draw the points only, without the line segments connecting them. If you need a graph
that is more precise, use numpoints increase the number of points which Maple uses to
create the plot (the default value could be 25 or 49 or 54). For example, the command
> plot(x^2*sin(x)-cos(3*x),x=0..Pi,numpoints=100);

will produce the plot of the function over [0, π], with the values of the function computed
at 100 points. Increasing numpoints is one way of obtaining a (more) precise plot of the
function. The other way is to “zoom-in” the graph, i.e. to plot it over a smaller interval.

It is possible to plot more functions in the same coordinate system (but it has to be
done over the same interval). The command is

plot({function,...,function},variable=interval)

For example, the command
> plot({x^2-x,3*x-2},x=-3..1);

will plot both x2 − x and 3x − 2 on the interval [−3, 1]. Unfortunately, Maple does not
label the graphs, and you should be able to determine which is which. The commands
> f := x -> sin(x)+2*cos(x);

and
> plot({f(x),f(x)+3,4*f(x)},x=-Pi..Pi);

will plot the functions f(x) = sin x + 2 cos x, f(x) + 3 = sin x + 2 cos x + 3 and 4f(x) =
4 sinx + 8 cosx on [−π, π], in the same coordinate system.

Similarly,
> plot({f(x),diff(f(x),x)},x=-Pi..Pi);

will give the plot of the function f(x) and its first derivative f ′(x).
Once the plot interval is selected, Maple chooses the range on the y-axis so that the

graph fits the window. But you can specify the range on the y-axis, if needed. The
command is

plot(function,variable=interval,range)

For example,
> plot(sin(x)+2,x=0..2*Pi,y=2..4);

or
> plot(sin(x)+2,x=0..2*Pi,y=0..3);

or
> plot(sin(x)+2,x=0..2*Pi,y=-1..5);

will plot the graphs of the same function, which appear different, due to different ranges
on the y-axis.

Solving Equations
The command that attempts to solve an equation is

solve(left side=right side,unknown)

if we ask for the algebraic (symbolic) expression as an answer, or

fsolve(left side=right side,unknown)

if we need a numeric approximation of the solution(s). For example,

> solve(a*x^2+b*x+c=0,x);

will return
−b ±

√
b2 − 4ac

2a
and > solve(x^2=2,x); will return

√
2 and −

√
2.

On the other hand, > fsolve(x^2=2,x); will respond with numeric approximations
1.414213562 and −1.414213562 of

√
2 and −

√
2. Sometimes, solve command will not

return anything. For example, > solve(cos(x)=x,x); will give no meaningful answer,
whereas > fsolve(cos(x)=x,x); gives the solution 0.7390851332 in the decimal form.

We can also tell Maple where to look for solutions. The command

fsolve(left side=right side,unknown=interval)

will look for the solution(s) in the given interval. For example,

> fsolve(cos(x)=0.1*x,x=0..1);

will give the solution of cosx = 0.1x that belongs to the interval [0, 1].
There are situations when Maple can not find all solutions to a given equation. If

this is the case, the combination of the graph of the function and fsolve commands with
specified interval (that is read off from the graph) might be very useful.

Differentiation
The command that computes the first derivative of a function is

diff(function,variable)

where variable denotes the variable with respect to which the derivative is computed. To
compute the derivative of sin(3 cosx) + 3.2 ln(x + 4) we use

> diff(sin(3*cos(x))+3.2*ln(x+4),x);

Remember that you can use simplify(%) to try to convert the result to a more useful
form. We can define the function first, and then compute its derivative; the commands

> F := u -> u^3+2*u^2-exp(u)+5;

and

> diff(F(u),u);

will return the derivative of F (u) = u3 + 2u2 − eu + 5. It does not suffice to type F for the
function: Maple needs F(u).

The command for the nth derivative of a function is

diff(function,variable$n)

For example, the commands

> f := x -> arcsin(x^2+x)+sin(cos(x));

and

> diff(f(x),x$3);

will define the function f(x) = arcsin(x2 +x)+sin(cos x) and compute its third derivative.
Derivative of a function is again a function, so all relevant commands apply. For

example, the command

> plot(diff(f(x),x),x=0..4);

will plot the derivative of the (previously defined) function f(x) over the interval [0, 4].
The result of the execution of

> plot({f(x),diff(f(x),x),diff(f(x),x$2)},x=-2..-1);

is the plot of the function f(x) and its first and second derivatives (on the same graph)
over the interval [−2,−1].

To evaluate the value of a derivative of a function, we use

subs(variable=value,function)

The command

> evalf(subs(x=1.5,diff(f(x),x$2)));

will return the numeric value (that is the reason why we added evalf) of the second
derivative of f(x) at x = 1.5. Similarly,

> subs(x=a, diff(f(x),x$3));

will give the symbolic expression that is the value of the third derivative of f(x) at x = a.
To compute the slope m of the tangent line to the graph of h(x) = arctanx at x = π, we
execute the command

> h := x -> arctan(x);

to define the function h(x), and then

> m := subs(x=Pi,diff(h(x),x));

Integration
To find an indefinite integral of a function, we execute the command

int(function,variable)

where function is the integrand and variable the variable of integration. For example, to
find ∫

(arctan x + ex + 7x − 2) dx,

we use
> int(arctan(x)+exp(x)+7*x-2,x);

The definite integral is computed similarly:

int(function,variable=interval)

where interval represents the interval of integration, in the form a..b (a is a real number
or −∞ (Maple’s symbol for ∞ is infinity), and b is a real number or ∞). For example,
to find ∫ 3.4

2
(t + 1.33 ln t) dt,

we use
> int(t+1.33*ln(t),t=2..3.4);

The command
> int(exp(-x^2),x=-infinity..infinity);

returns the value of the definite integral∫ ∞

−∞
e−x2

dx.

We can define the function first, and then integrate it. For example, to compute
∫ 5

1

x2 + 1 − x

3x4 − x2
dx,

we execute
> f := x -> (x^2+1-x)/(3*x^4-x^2);

and then
> evalf(int(f(x),x=1..5));

To define the function
g(x) =

∫ sin x

0
(t2 + 1) dt

we execute
> g := x -> int(t^2+1,t=0..sin(x));

Now we can do anything we do with any other function: for example, to evaluate its value
at x = π/2 we use
> evalf(g(Pi/2));

or
> evalf(subs(x=Pi/2,g(x)));

We can plot g(x)
> plot(g(x),x=0..2*Pi);

(we chose the plot interval to be [0, 2π]), compute its derivative
> diff(g(x),x);

or integrate it (say, over the interval [0, 1])
> evalf(int(g(x),x=0..1));

